

```

* Stata Analysis Examples Replication for ASDA 2nd Edition
* Berglund April 2017
* Chapter 12

use "P:\asda 2\data sets\nhanes 2011_2012\c12_impute_subset_nhanes1112.dta" if age18p==1 & wtmec2yr > 0 , clear

* Section 12.6 MI and FI methods using NHANES 2011-2012 data.
* examine missing data patterns.
gen weight=wtmec2yr
misstable patterns bpxdil_1 bmx bmi indfmpir marcat riagendr ridreth1 agec agecsq weight descodes if age18p==1
summarize
*****
* complete case analysis, numbers for tables 12.4 and 12.5
svyset sdmvpsu [pweight=wtmec2yr], strata(sdmvstra)

* high blood pressure indicator
gen high_diastolic = .
replace high_diastolic=0 if bpxdil_1 < 90
replace high_diastolic =1 if bpxdil_1 >=90 & bpxdil_1 < .

* Proportion with high diastolic blood pressure.
svy: prop high_diastolic
* Logistic regression
svy: logistic high_diastolic i.ridreth1 i.riagendr agec agecsq, coef

*****
* MI Method with Design Variables in Imputation Model
* set output data set to full long style
mi set flong

* set vars to be imputed
mi register imputed marcat bmx bmi indfmpir bpxdil_1

* set vars with fully observed data plus the str_cluster and weight variables
mi register regular riagendr ridreth1 agec descodes wtmec2yr

* mi impute chained using main effects plus descodes weight as predictors
mi impute chained (mlogit) marcat (regress) ///
bmx bmi (regress, include(i.descodes wtmec2yr)) ///
bpxdil_1 (regress) indfmpir=i.riagendr i.ridreth1 agec agecsq , add(5) rseed(2016) replace

* set survey variables
mi svyset sdmvpsu [pweight=wtmec2yr] , strata(sdmvstra)

* means analysis with imputed data : BMI, poverty index diastolic blood pressure.
mi estimate, noisily vartable : svy: mean bmx bmi indfmpir bpxdil_1

* high blood pressure indicator
gen high_diastolic1 = 0
replace high_diastolic1 =1 if bpxdil_1 >=90

* Proportion with high diastolic blood pressure.
mi estimate: svy: prop high_diastolic1

* logistic model with imputed data and outcome coded as bpxdil_1 > 90 = 1 and 0 otherwise.
mi estimate: svy: logistic high_diastolic1 i.ridreth1 i.riagendr agec agecsq
* joint test of gender and race/ethnicity from logistic model with design variables and weights in imputation
model.
mi test 2.riagendr 2.ridreth1 3.ridreth1 4.ridreth1 5.ridreth1

*****
* MI Method without Design Variables but with Weight in Imputation Model
use "P:\asda 2\data sets\nhanes 2011_2012\c12_impute_subset_nhanes1112.dta" if age18p==1 & wtmec2yr > 0 , clear
* set output data set to full long style
mi set flong

* set vars to be imputed
mi register imputed marcat bmx bmi indfmpir bpxdil_1

* set vars with fully observed data plus the str_cluster and weight variables
mi register regular riagendr ridreth1 agec descodes wtmec2yr

* mi impute chained using main effects but no design variables in model
mi impute chained (mlogit) marcat (regress) bmx bmi (regress) ///
bpxdil_1 (regress) indfmpir = i.riagendr i.ridreth1 agec agecsq [pweight=wtmec2yr], add(5) rseed(2016) replace

```

```
* set survey variables
mi svyset sdmvpsu [pweight=wtmec2yr] , strata(sdmvstra)

* means analysis with imputed data : BMI, poverty index diastolic blood pressure.
mi estimate, noisily vartable : svy: mean bmx bmi indfmpir bpxdi1_1

* high blood pressure indicator
gen high_diastolic2 = 0
replace high_diastolic2 =1 if bpxdi1_1 >=90

* Proportion with high diastolic blood pressure.
mi estimate: svy: prop high_diastolic2

* logistic model with imputed data and outcome coded as bpxdi1_1 > 90 = 1 and 0 otherwise.
mi estimate: svy: logistic high_diastolic2 i.ridreth1 i.riagendr agec agecsq
* test of gender and race/ethnicity from logistic model with design variables and weights in imputation model.
mi test 2.riagendr 2.ridreth1 3.ridreth1 4.ridreth1 5.ridreth1
```